

Tipps & Tricks: Table Functions

Bereich:	PL/SQL	Erstellung:	01/2005 HA
Versionsinfo:	9.2, 10.2, 11.1	Letzte Überarbeitung:	06/2009 EF

Table Functions

Eine table function ist eine PL/SQL-Funktion, die eine Gruppe von Datensätzen, eine [Collection](#) sowohl als Übergabeparameter aufnehmen als auch als Returnwert zurückgeben kann.

Im Beispiel unten wird an eine Table Function ein Ref Cursor übergeben, Rückgabewert ist ein Varray.

Beispiel:

```
CREATE TYPE dept_info AS TABLE OF VARCHAR2(1000);
/

CREATE OR REPLACE PACKAGE cursor_pack
IS
    TYPE ref_cur IS REF CURSOR;
END;
/

CREATE OR REPLACE FUNCTION dept_table_function
( p_cur IN cursor_pack.ref_cur )
RETURN dept_info IS
    v_table dept_info := dept_info();
    v_deptno dept.deptno%TYPE;
    v_dname dept.dname%TYPE;
    v_loc dept.loc%TYPE;
    i number := 0;
BEGIN
    LOOP
        i := i + 1;
        FETCH p_cur INTO v_deptno, v_dname, v_loc;
        EXIT WHEN p_cur%notfound;
        v_table.extend;
        v_table(i) :=
            'Abteilung ' || v_deptno || ' ( ' || v_dname || ' ) befindet sich in ' || v_loc;
    END LOOP;
    CLOSE p_cur;
    RETURN v_table;
END;
/
```

Eine Table Function kann man mit dem TABLE-Keyword in der FROM-Klausel eines SELECT aufrufen.

```
SELECT * FROM TABLE(dept_table_function(CURSOR(SELECT deptno, dname,
loc FROM dept)));
```

COLUMN_VALUE

Abteilung 10 (ACCOUNTING) befindet sich in NEW YORK
Abteilung 20 (RESEARCH) befindet sich in DALLAS
Abteilung 30 (SALES) befindet sich in CHICAGO
Abteilung 40 (OPERATIONS) befindet sich in BOSTON